

BASIC 语言应用程序100例

周丽芝 吴 林 钱乙余 编译

新 时 代 出 版 社

内 容 简 介

本书是 BASIC 语言应用程序汇编, 它既为编程初学者提供各类程序的范例, 也是一个 BASIC 语言的应用程序库, 可供程序设计人员参考。

书中汇编了 100 个 BASIC 应用程序, 其中有关科学和工程问题计算和分析的应用程序 72 个, 字符串处理应用程序 17 个, 图形显示的应用程序 11 个。书中不仅给出了每个实例的具体程序, 而且都附有程序框图和简要的说明。

本书可供各领域中应用计算机的工程技术人员参考, 也可供计算机初学者或中学生学习。

JS/24/18

BASIC 语言应用程序 100 例

周丽芝 吴 林 钱乙余 编译

新时代出版社出版 新华书店北京发行所发行

国防工业出版社印刷厂印刷

787×1092毫米 32开本 8.125印张 177千字

1986年6月第1版 1986年6月北京第1次印刷

印数: 00001—17800册

统一书号: 15241·69 定价: 1.70元

前 言

本书是BASIC语言应用程序的汇编。它将为计算机爱好者和初学者学习自行编制BASIC语言程序提供各种范例,也可供计算机软件工作者启发思路和参考。

本书共汇编了BASIC应用程序100例,取材力求具有实用、内容新颖、能开阔思路和提高兴趣等特点。其内容包括:计算分析、字符串处理、图象显示等三个方面。大部分实例都是各类科学和工程问题的实际计算方法、数据处理和图象显示。也选编了少数游戏,它既可以提高学习兴趣,也有益于启发思路。

每个实例都给出了具体的程序,并附有程序框图和简要的说明。程序编排的顺序,尽力采取由浅入深、从简单到复杂,以便于学习和掌握它。

书中汇编的100个程序,都在日本SHARPMZ-80和美国APPLE-Ⅱ微计算机上进行了运行校核,因此,如果使用上述微计算机或相同类型的微计算机,可以根据需要直接选用。如果使用不同于上述类型的微计算机,需要参考所用微计算机的BASIC语言使用手册,作必要的语句性修改,就可以使用。

为了方便,书后附有上述微计算机BASIC程序设计语言的语句、指令、函数运算等一览表,以备检索和对照。

由于编译者水平有限,书中一定有许多错误或不当之处,敬请批评指正。

目 录

一、计算分析	1
1. 几何平均值	1
2. 几何级数	2
3. 调和平均值	5
4. 调和数	6
5. 调和级数	8
6. 算术平均值	11
7. 算术级数	13
8. 综合平均值	15
9. 费保纳赛数 (Fibonacci Number)	17
10. 最大公约数	20
11. 质数判别	21
12. 随机数组中的最大数和最小数	23
13. 随机数组中数的有序排列	24
14. 函数的最小值	26
15. 函数的最大值	27
16. 线性插值	29
17. 直角坐标变极坐标	31
18. 极坐标变直角坐标	34
19. 坐标变换和/或旋转	36
20. 三点确定圆	38
21. 圆周上的点	40
22. 任意底对数	43
23. 向量叉乘积	44

24. 向量点乘积与模	46
25. 自然对数的底	48
26. 正弦函数	49
27. 正切函数	51
28. 双曲线函数	53
29. 反双曲线函数	56
30. 多项式	59
31. 阶乘	60
32. 排列与组合	62
33. Γ 函数及阶乘	64
34. 一元二次方程	67
35. 代数方程 (一)	69
36. 代数方程 (二)	72
37. 2×2 矩阵的行列式及逆矩阵	74
38. 3×3 矩阵的行列式及逆矩阵	76
39. 二元一次方程组	79
40. 三元一次方程组	81
41. 线性方程组 (一)	84
42. 线性方程组 (二)	87
43. 一阶导数	92
44. 常微分方程 (一)	94
45. 常微分方程 (二)	97
46. 定积分	99
47. 数值积分 (一)	101
48. 数值积分 (二)	104
49. 数值积分 (三)	106
50. X^2 (CHI-SQUARE) 值	108
51. 概率	110
52. 非重复概率	112
53. 高斯概率函数	114

54. 矩、偏倚系数及峰度	116
55. 平均值、标准偏差和标准误差	119
56. 圆形截面惯性矩、极惯性矩与面积	121
57. 环形截面惯性矩、极惯性矩与面积	122
58. 矩形截面惯性矩、极惯性矩与面积	124
59. 最小损耗 L 衰减器	126
60. 电阻衰减器	128
61. Π 网络电抗匹配	132
62. 绘制曲线	134
63. 打印正弦曲线	137
64. 单利和复利	139
65. 定额零存整取	141
66. 复利综合计算	144
67. 直线折旧	147
68. 猜数	149
69. 求三角形面积	151
70. 试作直角三角形	154
71. 聪明的小童工	156
72. 重新排号	157
二、字符串处理	160
1. “好友”	160
2. “BASIC 语言”	160
3. 数值字符变换	161
4. 两位数加法	162
5. 小数点对位	163
6. 整数定位分隔	164
7. 课程表	166
8. 成绩统计及名次排列	167
9. 工资表 (一)	170
10. 工资表 (二)	172

11. 编制目录	174
12. 电话册	177
13. 世界钟	179
14. 时间显示	182
15. 春装	189
16. 分发牌与洗牌	192
17. 机动电话册	195
三、图形显示	209
1. 矩形	209
2. 斜线	210
3. 波纹	211
4. 球的运动	212
5. 帆船	213
6. 兔与狐	215
7. 占位置	218
8. 花纹图案	220
9. 色彩及彩色图案	222
10. 赛球	226
11. 打靶	230
附录一 MZ-80 BASIC 语言一览表	241
附录二 APPLE-Ⅱ 部分扩展 BASIC 语言一览表	251
参考资料	252

一、计算分析

1. 几何平均值

本程序根据使用者给出的取样值计算其几何平均值。连续运算到给出的取样值为零时，计算停止。

公式：

$$G = \sqrt[n]{(a_1)(a_2)\cdots(a_n)}$$

式中 G ——几何平均值；

$a_1, a_2, \dots, a_n$ ——取样值；

n ——取样数。

框图如图 1-1 所示。




图 1-1
8610741

程序:

```

10  REM THIS PROGRAM COMPUTES THE GEOMETRIC MEAN
20  LET Y = 1
30  LET N = 0
40  PRINT "A = ",
50  INPUT A
60  IF A = 0 THEN 130
70  LET N = N + 1
80  LET Y = Y * A
90  LET G = Y ↑ (1/N)
100 PRINT "N = ", N, "A = ", A
110 PRINT "G = ", G
120 GOTO 40
130 PRINT
140 PRINT "TYPE 1 TO CONTINUE, 0 TO STOP"
150 INPUT L
160 IF L = 1 THEN 180
170 STOP
180 GOTO 20
190 END

```

输入: A 1 = 2 N = 1

计算结果: G = 2

输入: A 2 = 5 N = 2

计算结果: G = 3.1623

2. 几何级数

本程序根据首项、公比值及项数计算几何级数与级数之和。

公式:

$$A, AR, AR^2, \dots, AR^{N-1}$$

$$S = A + AR + AR^2 + \dots + AR^{N-1}$$

式中 A——首项;

100100

R——公比；
N——项数；
S——级数和。

框图如图 1-2 所示。




图 1-2

程序：

```

10  REM THIS PROGRAM COMPUTES THE VALUES
20  REM AND THEIR SUM OF A GEOMATRIC PROGRESSION
30  PRINT "A = ";

```

```

40  INPUT A
50  PRINT "R =",
60  INPUT R
70  PRINT "N =",
80  INPUT N
90  PRINT "FOR TABLE TYPE 1, IF NOT TYPE 0"
100 INPUT C
110 IF C = 1 THEN 130
120 GOTO 160
130 PRINT
140 PRINT "GEOMETRIC PROGRESSION"
150 PRINT "TERM NUMBER", "TERM VALUE",
160 LET J = 0
170 FOR I = 0 TO N - 1
180 LET K = I + 1
190 LET L = A * (R ↑ I)
200 LET J = J + L
210 IF C = 1 THEN 230
220 GOTO 240
230 PRINT K, L
240 NEXT I
250 PRINT "S =", J
260 PRINT
270 PRINT "TYPE 1 TO CONTINUE, 0 TO STOP"
280 INPUT X
290 IF X = 1 THEN 310
300 STOP
310 PRINT
320 GOTO 30
330 END

```

输入: $A=10$ $R=2$ $N=5$

计算结果: 几何级数

项号	项值
1	10
2	20

3	40
4	80
5	160

$$S = 310$$

3. 调和平均值

根据使用者给定的取样值，本程序计算调和平均值。当给定的取样值为零时，运算停止。

公式：

$$H = \frac{N}{\sum_{i=1}^N \frac{1}{a_i}}$$

式中 H——调和平均值；

a_i ——取样值；

N——取样数。

框图如图 1-3 所示。

程序

```

10  REM THIS PROGRAM COMPUTES HARMONIC MEAN
20  LET Z = 0
30  LET N = 0
40  PRINT "A = ";
50  INPUT A
60  IF A = 0 THEN 120
70  LET N = N + 1
80  LET Z = Z + (1 / A)
90  LET H = N / Z
100 PRINT "N =", N, "A =", A, "H =", H
110 GOTO 40
120 PRINT
130 PRINT "TYPE 1 TO CONTINUE, 0 TO STOP"
```

```

140  INPUT L
150  IF L = 1 THEN 170
160  STOP
170  PRINT
180  GOTO 20
190  END

```




图 1-3

输入: $N = 1$ $A(a_1) = 2$

计算结果: $H = 2$

输入: $N = 2$ $A(a_2) = 5$

计算结果: $H = 2.86$

输入: $N = 3$ $A(a_3) = 7$

计算结果: $H = 3.56$

4. 调和数

本程序计算 N 项调和数, N 由使用者给出。

公式:

1, $1 + 1/2$, $1 + 1/2 + 1/3$, $1 + 1/2 + 1/3 + 1/4$, ...
框图如图 1-4 所示。




图 1-4

程序

```

10  REM THIS PROGRAM COMPUTES HARMONIC NUMBERS
20  PRINT "N =",
30  INPUT N
40  LET K = 0
50  LET D = 0
60  PRINT "TERM NUMBER", "TERM VALUE"
70  FOR I = 0 TO N
80  LET K = I + 1
90  LET C = 1/K
100 LET D = D + C
110 PRINT K, D
120 NEXT I
  
```

130 PRINT

140 END

输入: $N=20$

计算结果:

项号(K)	项值(D)
1	1
2	1.5
3	1.83333
4	2.08333
5	2.28333
6	2.45
7	2.59285
8	2.71785
9	2.82896
10	2.92986
11	3.01987
12	3.10321
13	3.18013
14	3.25156
15	3.31822
16	3.38072
17	3.43955
18	3.49510
19	3.54774
20	3.59773
21	3.64535

5. 调和级数

本程序计算调和级数之和及项值,由使用者给定A、B,差D及项数N。

公式:

$$\frac{A}{B}, \frac{A}{B+D}, \frac{A}{B+2D}, \dots, \frac{A}{B+(N-1)D}$$

$$S = \frac{A}{B} + \frac{A}{B+D} + \frac{A}{B+2D} + \cdots + \frac{A}{B+(N-1)D}$$

式中 S ——级数和。

框图如图 1-5 所示。




图 1-5

程序:

```

10  REM THIS PROGRAM COMPUTES A SERIES OF HARMONIC
    PROGRESSIONS
20  PRINT "A =",
30  INPUT A
40  PRINT "B =",
50  INPUT B
60  PRINT "D =",
70  INPUT D
80  PRINT "N =",
90  INPUT N
100 PRINT "FOR TABLE TYPE 1, IF NOT TYPE 0"
110 INPUT C
120 IF C = 1 THEN 140
130 GOTO 160
140 PRINT "HARMONIC PROGRESSION"
150 PRINT "TERM NUMBER", "TERM VALUE"
160 LET J = 0
170 FOR I = 0 TO N - 1
180 LET K = I + 1
190 LET L = A / (B + (I * D))
200 LET J = J + L
210 IF C = 1 THEN 230
220 GOTO 240
230 PRINT K, L
240 NEXT I
250 PRINT "S =", J
260 PRINT
270 PRINT "TYPE 1 TO CONTINUE, 0 TO STOP"
280 INPUT X
290 IF X = 1 THEN 310
300 STOP
310 PRINT
320 GOTO 20
330 END

```